

## 17.2. CÁC HÀM THÔNG DỤNG

### 17.2.1. Các hàm toán học (Math & Trig)

Cú pháp	Ý nghĩa và ví dụ
<code>ABS(<i>number</i>)</code>	Trả về giá trị tuyệt đối của một số thực. =ABS(12 - 20) → 8
<code>INT(<i>number</i>)</code>	Trả về số nguyên lớn nhất không vượt quá <i>number</i> . =INT(5.6) → 5 =INT(-5.6) → -6
<code>MOD(<i>number</i>, <i>divisor</i>)</code>	Trả về số dư của phép chia nguyên <i>number</i> cho <i>divisor</i> ( <i>number</i> , <i>divisor</i> là các số nguyên). =MOD(5, 3) → 2
<code>ODD(<i>number</i>)</code>	Làm tròn trên tới một số nguyên lẻ gần nhất. =ODD(3.6) → 5 =ODD(-2.2) → -3
<code>PRODUCT(<i>number1</i>, <i>number2</i>, ...)</code>	Tính tích của các giá trị trong danh sách tham số. =PRODUCT(2, -6, 3, 4) → -144
<code>RAND()</code>	Trả về một số ngẫu nhiên trong khoảng từ 0 đến 1. =RAND() → Số ngẫu nhiên
<code>ROUND(<i>number</i>, <i>num_digits</i>)</code>	Làm tròn số <i>number</i> với độ chính xác đến <i>num_digits</i> chữ số thập phân (với qui ước 0 là làm tròn tới hàng đơn vị, 1 là lấy 1 chữ số thập phân, -1 là làm tròn tới hàng chục, ...). =ROUND(5.13687, 2) → 5.14 =ROUND(145.13687, -2) → 100
<code>SQRT(<i>number</i>)</code>	Tính căn bậc 2 của một số dương <i>number</i> . =SQRT(36) → 6
<code>SUM(<i>number1</i>, <i>number2</i>, ...)</code>	Tính tổng của các giá trị trong danh sách tham số. =SUM(2, -6, 8, 4) → 8
<code>SUMIF(<i>range</i>, <i>criteria</i> [, <i>sum_range</i>])</code>	Tính tổng các ô thỏa mãn điều kiện. - <b>range</b> : vùng mà điều kiện sẽ được so sánh. - <b>criteria</b> : chuỗi mô tả điều kiện. Ví dụ: "10", ">15", "<20", ... - <b>sum_range</b> : vùng được tính tổng. Các ô trong vùng này sẽ được tính tổng nếu các ô tương ứng trong vùng <i>range</i> thỏa điều kiện. Nếu không có <i>sum_range</i> thì vùng <i>range</i> sẽ được tính. =SUMIF(C4:C12, ">=6", F4:F12) =SUMIF(C4:C12, ">=6") =SUMIF(B4:B12, "NV", G4:G12)

### 17.2.2. Các hàm thống kê (Statistical)

Cú pháp	Ý nghĩa và ví dụ
MAX( <i>number1, number2, ...</i> )	Trả về giá trị lớn nhất của các giá trị số trong danh sách tham số. =MAX(1, 2, 3, 5) → 5
MIN( <i>number1, number2, ...</i> )	Trả về giá trị nhỏ nhất của các giá trị số trong danh sách tham số. =MIN(1, 2, 3, 5) → 1
AVERAGE( <i>number1, number2, ...</i> )	Trả về giá trị trung bình cộng của các số trong danh sách tham số. =AVERAGE(1, 2, 3, 5) → 2.75
COUNT( <i>value1, value2, ...</i> )	Đếm số các <b>giá trị số</b> trong danh sách tham số. =COUNT(2, “hai”, 4, -6) → 3
COUNTA( <i>value1, value2, ...</i> )	Đếm số các <b>ô không rỗng</b> trong danh sách tham số. =COUNT(2, “hai”, 4, -6) → 4
COUNTBLANK( <i>range</i> )	Đếm số các <b>rỗng</b> trong vùng <i>range</i> . =COUNTBLANK(B4:B12)
COUNTIF( <i>range, criteria</i> )	Đếm các ô thỏa mãn điều kiện <i>criteria</i> trong vùng <i>range</i> . - <b>range</b> : là vùng mà điều kiện sẽ được so sánh. - <b>criteria</b> : là chuỗi mô tả điều kiện. Ví dụ: "10", ">15", "<20". =COUNTIF(B4:B12, ">=6")
RANK( <i>number, ref [, order]</i> )	Trả về thứ hạng của <i>number</i> trong <i>ref</i> , với <i>order</i> là cách xếp hạng. Nếu order = 0 hoặc được bỏ qua thì <i>ref</i> được hiểu là có thứ tự giảm. Nếu order <> 0 thì <i>ref</i> được hiểu là có thứ tự tăng. =RANK(F4, \$F\$4:\$F\$12, 0) =RANK(G4, \$G\$4:\$G\$12, 1)

### 17.2.3. Các hàm Logic (Logical)

Cú pháp	Ý nghĩa và ví dụ
AND( <i>logical1, logical2, ...</i> )	Trả về giá trị TRUE nếu tất cả các điều kiện đều là TRUE. =AND(3>2, 5<8, 9>-12) → TRUE
OR( <i>logical1, logical2, ...</i> )	Trả về giá trị TRUE nếu có ít nhất một điều kiện là TRUE. =OR(2>3, 12<8, 9>3) → TRUE =OR(2>3, 12<8, -9>3) → FALSE
NOT( <i>logical</i> )	Lấy phủ định của giá trị <i>logical</i> . =NOT(2>3) → TRUE

<p>IF(logical_test, value_if_true, value_if_false)</p>	<p>Trả về giá trị thứ nhất <i>value_if_true</i> nếu điều kiện <i>logical_test</i> là TRUE, ngược lại sẽ trả về giá trị thứ hai <i>value_if_false</i>.</p> <p>=IF(A1 &gt;=5, "Đậu", "Rớt")</p> <p>Nếu giá trị tại A1 &gt;= 5 thì kết quả của hàm là Đậu. Ngược lại nếu giá trị ở ô A1 &lt; 5 thì kết quả là Rớt.</p>
--	---

#### 17.2.4. Các hàm xử lý chuỗi (Text)

Cú pháp	Ý nghĩa và ví dụ
LOWER(text)	Chuyển chuỗi <i>text</i> thành chữ thường. =LOWER("Đại học CAN Tho") → đại học can tho
UPPER(text)	Chuyển chuỗi <i>text</i> thành chữ in hoa. =UPPER("Đại học CAN Tho") → DAI HOC CAN THO
PROPER(text)	Đổi các ký tự đầu của mỗi từ trong chuỗi <i>text</i> thành chữ in hoa, còn lại đều là chữ thường. =PROPER("Đại học CAN Tho") → Dai Hoc Can Tho
TRIM(text)	Cắt bỏ các ký tự trống vô ích trong chuỗi <i>text</i> . =TRIM(" Can Tho ") → Can Tho
LEN(text)	Trả về độ dài của chuỗi <i>text</i> (số ký tự trong chuỗi <i>text</i> ). =LEN("Đại học CAN Tho") → 15
LEFT(text, num_chars)	Trả về <i>num_char</i> ký tự bên trái chuỗi <i>text</i> . =LEFT("Đại học CAN Tho", 7) → Dai học
RIGHT(text, num_chars)	Trả về <i>num_char</i> ký tự bên phải chuỗi <i>text</i> . =RIGHT("Đại học CAN Tho", 7) → CAN Tho
MID(text, start_num, num_chars)	Trả về chuỗi ký tự có độ dài <i>num_chars</i> bắt đầu từ vị trí <i>start_num</i> của chuỗi <i>text</i> . =MID("Đại học CAN Tho", 5, 3) → học
VALUE(text)	Chuyển chuỗi có dạng số thành trị số. = VALUE("123") + 2 → 125
FIND(find_text, within_text [, start_num])	<p>Trả về vị trí xuất hiện (nếu có) của <i>find_text</i> trong <i>within_text</i> (bắt đầu tìm từ vị trí <i>start_num</i>).</p> <p><b>Chú ý:</b></p> <ul style="list-style-type: none"> <li>- Nếu không có <i>start_num</i> thì vị trí bắt đầu tìm từ đầu chuỗi.</li> <li>- Hàm FIND phân biệt chữ in hoa và chữ thường.</li> <li>- Nếu không tìm thấy <i>find_text</i> thì sẽ trả về lỗi #VALUE!</li> </ul> <p>=FIND("Excel", "Microsoft Excel") → 11 =FIND("Excel", "Microsoft Excel", 6) → 11 =FIND("excel", "Microsoft Excel", 6) → #VALUE!</p>
SEARCH(find_text, within_text [, start_num])	<p>Tương tự như hàm <b>FIND</b> nhưng không phân biệt chữ in hoa hay thường.</p> <p>=SEARCH("Excel", "Microsoft Excel") → 11 =SEARCH("excel", "Microsoft Excel") → 11</p>

REPLACE( <i>old_text</i> , <i>num_start</i> , <i>num_chars</i> , <i>new_text</i> )	Thay thế <i>num_chars</i> ký tự trong <i>old_text</i> bằng <i>new_text</i> bắt đầu từ vị trí <i>num_start</i> . =REPLACE(“Ngon ngu lap trinh”, 10, 3, “chuong”) → Ngon ngu chuong trinh
--	---

### 17.2.5. Các hàm ngày và giờ (Date & Time)

Giả sử ô A1 chứa ngày 28/09/2004 (Thứ ba).

Cú pháp	Ý nghĩa và ví dụ
TODAY( )	Trả về ngày hiện hành của hệ thống. =TODAY( ) → Tuỳ vào ngày hiện hành của hệ thống.
NOW( )	Trả về ngày và giờ hiện hành của hệ thống. =NOW( ) → Tuỳ vào ngày và giờ hiện hành của hệ thống.
DAY( <i>date</i> )	Trả về giá trị ngày trong tháng của biểu thức ngày <i>date</i> . =DAY(A1) → 28
MONTH( <i>date</i> )	Trả về giá trị tháng trong năm của biểu thức ngày <i>date</i> . =MONTH(A1) → 9
YEAR( <i>date</i> )	Trả về giá trị năm của biểu thức ngày <i>date</i> . =YEAR(A1) → 2004
WEEKDAY( <i>date</i> )	Trả về số thứ tự ngày trong tuần của biểu thức <i>date</i> . Giá trị 1: Sunday, 2:Monday, ..., 7: Saturday. =WEEKDAY(A1) → 3
DATEVALUE( <i>date_text</i> )	Đổi chuỗi ngày <i>date_text</i> (theo qui ước nhập ngày) thành trị số ngày. <b>Ghi chú:</b> ta có thể định dạng kết quả trên thành dạng Date bằng cách sử dụng menu <i>Format/Cells</i> . = DATEVALUE("22/8/55") → 20323 → 22/8/55
DATE( <i>year</i> , <i>month</i> , <i>day</i> )	Trả về giá trị dạng <i>Date</i> theo quy định của hệ thống. =DATE(2004,09,28) → 28/09/2004 =DATE(04,9,28) → 28/09/2004
TIME( <i>hour</i> , <i>minute</i> , <i>second</i> )	Trả về giá trị dạng <i>Time</i> . =TIME(8,25,28) → 8:25:28 AM =TIME(17,2,46) → 5:2:46 PM

### 17.2.6. Các hàm tìm kiếm (Lookup & Reference)

- **VLOOKUP**(*lookup\_value*, *table\_array*, *col\_index\_num*, *range\_lookup*)

Tìm giá trị *lookup\_value* trong cột trái nhất của bảng *table\_array* theo chuẩn dò tìm *range\_lookup*, trả về trị tương ứng trong cột thứ *col\_index\_num* (nếu tìm thấy).

*range\_lookup* = 1 (mặc nhiên):

Tìm tương đối, danh sách các giá trị dò tìm của bảng *table\_array* phải sắp xếp theo thứ tự tăng dần

Nếu tìm không thấy sẽ trả về giá trị lớn nhất nhưng nhỏ hơn *lookup\_value*.

*range\_lookup* = 0:

Tìm chính xác, danh sách các giá trị dò tìm của bảng *table\_array* không cần sắp xếp thứ tự.

Nếu tìm không thấy sẽ trả về lỗi #N/A.

- **HLOOKUP**(*lookup\_value*, *table\_array*, *row\_index\_num*, *range\_lookup*)

Tương tự như hàm VLOOKUP nhưng tìm giá trị *lookup\_value* trong dòng trên cùng của bảng *table\_array* theo chuẩn dò tìm *range\_lookup*, trả về trị tương ứng trong dòng thứ *row\_index\_num* (nếu tìm thấy)

**Ví dụ:** Cho bảng tính với số liệu như sau:

	A	B	C	D	E	F
1	A01	5		12	16	10
2	C02	6		15	20	24
3	B75	8		25	22	18
4						
5	A02	10		A01	B75	D25
6	B555	12		CÀ	THO	GẠO
7	D25	15		TRẮNG	NƯỚC	TRONG

=VLOOKUP("B75", A1:B3, 2, 0) → 8	=HLOOKUP(16, D1:F3, 3, 0) → 22
=VLOOKUP("B8", A1:B3, 2, 0) → #N/A	=HLOOKUP(15, D1:F3, 3, 0) → #N/A
=VLOOKUP("B85", A1:B3, 2, 1) → 5	=HLOOKUP(15, D1:F3, 3, 1) → 25
=VLOOKUP("B85", A1:B3, 2) → 5	=HLOOKUP(15, D1:F3, 3) → 25
=VLOOKUP(A6, A5:B7, 2, 0) → 12	=HLOOKUP(F5, D5:F7, 2, 0) → GẠO
=VLOOKUP("B555", A5:B7, 2, 0) → 12	=HLOOKUP("B75", D5:F7, 3, 1) → NƯỚC
=VLOOKUP("B85", A5:B7, 2, 1) → 12	=HLOOKUP("E95", D5:F7, 2, 0) → #N/A
=VLOOKUP("E05", A5:B7, 2) → 15	=HLOOKUP("E95", D5:F7, 3) → TRONG

- **MATCH**(*lookup\_value*, *lookup\_array*, *match\_type*): trả về vị trí (nếu tìm được) của *lookup\_value* trong mảng *lookup\_array* theo cách tìm *match\_type*

*match\_type* = 1:

Tìm tương đối, danh sách các giá trị dò tìm của bảng *table\_array* phải sắp xếp theo thứ tự tăng dần

Nếu tìm không thấy sẽ trả về vị trí của giá trị lớn nhất nhưng nhỏ hơn *lookup\_value*

*match\_type* = 0:

Tìm chính xác, danh sách các giá trị dò tìm của bảng *table\_array* không cần sắp xếp thứ tự

Nếu tìm không thấy sẽ trả về lỗi #N/A

*match\_type* = -1:

Tìm tương đối, danh sách phải sắp xếp các giá trị dò tìm của bảng *table\_array* theo thứ tự giảm dần

Nếu tìm không thấy sẽ trả về vị trí của giá trị nhỏ nhất nhưng lớn hơn *lookup\_value*

**Ví dụ:** sử dụng bảng dữ liệu ở phần ví dụ hàm VLOOKUP và HLOOKUP

= MATCH(16, D1:F1, 0) → 2	= MATCH(20, D2:F2, 0) → 2
= MATCH(18, D1:F1, 0) → #N/A	= MATCH(22, D2:F2, 1) → 2
= MATCH(15, D1:F1, 1) → 1	= MATCH(24, D3:F3, -1) → 3

- **INDEX(array, row\_num, column\_num):** trả về giá trị của ô ở hàng thứ *row\_num*, cột thứ *column\_num* trong mảng *array*.

**Ví dụ:** sử dụng bảng dữ liệu ở phần ví dụ hàm VLOOKUP và HLOOKUP

= INDEX(D1:F3, 2, 3) → 20

= INDEX(D1:F3, 4, 3) → #REF!

= INDEX(D1:F3, MATCH(26, D1:D3, 1), MATCH(16, D1:F1, 0)) → 22

### 17.2.7. Các hàm thông tin (ISfunction)

Các hàm thông tin dùng để kiểm tra xem kiểu của một giá trị hay của một ô có thỏa mãn một điều kiện nào đó không. Chẳng hạn: ô dữ liệu có phải là giá trị số không? Có phải là chuỗi ký tự không? ...

Các hàm thông tin luôn trả về một trong hai giá trị TRUE hoặc FALSE. Như vậy các hàm này có thể đáp ứng được trong các trường hợp mà có một số dữ liệu ngoại lệ trong một bảng dữ liệu cần tính toán.

**ISBLANK(value):** trả về giá trị TRUE nếu *value* là giá trị rỗng (blank), ngược lại thì trả về giá trị FALSE

**ISERROR(value):** trả về giá trị TRUE nếu *value* là một lỗi bất kỳ, ngược lại thì trả về giá trị FALSE.

**ISLOGICAL(value):** trả về giá trị TRUE nếu *value* là một giá trị logic, ngược lại thì trả về giá trị FALSE.

**ISNA(value):** trả về giá trị TRUE nếu *value* là lỗi #N/A, ngược lại thì trả về giá trị FALSE.

**ISNUMBER(value):** trả về giá trị TRUE nếu *value* là giá trị số, ngược lại thì trả về giá trị FALSE.

**ISTEXT(value):** trả về giá trị TRUE nếu *value* là một chuỗi, ngược lại thì trả về giá trị FALSE.

**Ví dụ:** Cho bảng tính với số liệu như sau:

	A	B	C
1	MACB	LCB	
2	111	333	
3	112	444	
4	113	555	

= ISBLANK(C1)	→ TRUE
= ISBLANK(A1)	→ FALSE
= ISERROR(MOD(114,0))	→ TRUE
= ISERROR(MOD(114,3))	→ FALSE
= ISLOGICAL(2>3)	→ TRUE
= ISERROR(VLOOKUP(114,A2:B4,2,FALSE))	→ TRUE
= ISNA(VLOOKUP(114,A2:B4,2,FALSE))	→ TRUE
= ISNUMBER(12345)	→ TRUE
= ISNUMBER("12345")	→ FALSE
= ISTEXT(A1)	→ TRUE
= ISTEXT(VALUE("12345"))	→ FALSE

#### 17.2.8. Ví dụ về cách sử dụng hàm

- Hàm IF(logical\_test, value\_if\_true, value\_if\_false)

##### Ví dụ 1:

=IF(B1 >= 5, "Đậu", "Rớt")

Excel sẽ kiểm tra biểu thức B1 >= 5, nếu biểu thức đúng (giá trị tại ô B1 là >= 5) thì sẽ in ra "Đậu" và kết thúc hàm, ngược lại sẽ in ra "Rớt" và kết thúc hàm.

##### Ví dụ 2:

=IF(B1 > 0, "Số dương", IF(B1 = 0, "Số không", "Số âm"))

- Excel sẽ kiểm tra biểu thức B1 > 0, nếu biểu thức đúng thì sẽ in ra "Số dương" và kết thúc hàm, ngược lại sẽ xét tiếp biểu thức B1 = 0.

- Nếu biểu thức B1 = 0 là đúng thì sẽ in ra "Số không" và kết thúc hàm, ngược lại sẽ in ra "Số âm" và kết thúc hàm.

**Ví dụ 3:** giả sử yêu cầu xếp loại học tập dựa vào Diem TB trong bảng điểm cho trước và cách xếp loại như sau:

Nếu Diem TB >= 9 → XS

Nếu 8 <= Diem TB < 9 → Giỏi

Nếu 7 <= Diem TB < 8 → Khá

Nếu 5 <= Diem TB < 7 → TB

Nếu 3.5 <= Diem TB < 5 → Yếu

Nếu Diem TB < 3.5 → Kém

	A	B	C	D
1	STT	Ten	Diem TB	Xep loai
2		Cần	6.7	
3		Kiểm	9.2	
4		Liên	5.8	
5		Chính	2.4	
6		Chí	7.7	
7		Công	7.9	

- Công thức tại ô D2:

=IF(C2 >= 9, "XS", IF(C2 >= 8, "Giỏi", IF(C2 >= 7, "Khá", IF(C2 >= 5, "TB", IF(C2 >= 3.5, "Yếu", "Kém")))))

- Sao chép công thức tại ô D2 đến vùng D3:D7

**Tổng quát:** nếu có  $n$  trường hợp thì ta phải sử dụng  $n-1$  hàm IF lồng nhau.

- Hàm **VLOOKUP**(lookup\_value, table\_array, row\_index\_num, range\_lookup)

**Ví dụ:** Cho dữ liệu như bảng dưới đây:

	A	B	C	D	E	F	G
1			<b>BẢNG HỌC BỔNG</b>			<b>BẢNG TRỢ CẤP</b>	
2			<b>Xếp loại</b>	<b>Học bổng</b>		<b>Mã TC</b>	<b>Tỉ lệ</b>
3			01	100,000		B	50%
4			02	70,000		A	100%
5			03	50,000		C	0%
6			04	30,000			
7							
8	<b>DANH SÁCH NHẬN TIỀN HỌC BỔNG</b>						
9							
10	<b>TT</b>	<b>Họ tên</b>	<b>Xếp loại</b>	<b>Học bổng</b>	<b>Mã TC</b>	<b>Trợ cấp</b>	<b>Tổng cộng</b>
11		Trường	02	70,000	A	70,000	
12		Kỳ	01		B		
13		Kháng	02		C		
14		Chiến	04		B		
15		Nhất	01		C		
16		Định	03		B		
17		Thắng	04		A		
18		Lợi	02		A		

**Yêu cầu:**

1) Tính cột **Học bổng** dựa vào cột **Xếp loại** và **BẢNG HỌC BỔNG**.

2) Tính cột **Trợ cấp** = **Học bổng** \* **Tỉ lệ**

Trong đó **Tỉ lệ** được tính nhờ vào cột **Mã TC** và **BẢNG TRỢ CẤP**.

3) Tính cột **Tổng cộng** = **Học bổng** + **Trợ cấp**

**Giải:**

1) Tính cột **Học bổng**

+ Trước hết ta viết công thức cho ô D11:

Lấy giá trị trong ô C11 (lookup\_value) để dò trong vùng \$C\$3:\$D\$6 (table\_array), trong bảng này ta muốn lấy cột Học bổng tức là cột thứ 2 (col\_index\_num), do trong BẢNG HỌC BỔNG cột Xếp loại đã sắp xếp theo thứ tự tăng dần nên ta có thể dò tìm tương đối (range\_lookup là 1 hoặc có thể bỏ qua).



Vì vùng C3:D6 sử dụng chung để dò tìm nên phải lấy địa chỉ tuyệt đối. Ta được công thức cho ô D11 như sau:

=VLOOKUP(C11, \$C\$3:\$D\$6, 2, 1)

hoặc =VLOOKUP(C11, \$C\$3:\$D\$6, 2)

+ Sao chép công thức tại ô D11 đến vùng D12:D18.

## 2) Tính cột **Trợ cấp**

+ Trước hết ta viết công thức cho ô F11:

Để tính Tỷ lệ ta lấy giá trị trong ô E11 (*lookup\_value*) để dò trong vùng \$F\$3:\$G\$5 (*table\_array*), trong bảng này ta muốn lấy cột Tỷ lệ tức là cột thứ 2 (*col\_index\_num*), do trong BẢNG TRỢ CẤP cột Mã TC chưa được sắp xếp nên ta phải dò tìm tuyệt đối (*range\_lookup* là 0).

Ta được công thức cho ô F11 như sau:

=D11 \* VLOOKUP(E11, \$F\$3:\$G\$5, 2, 0)

+ Sao chép công thức tại ô F11 đến vùng F12:F18.

## 3) Tính cột **Tổng cộng**

+ Trước hết ta viết công thức cho ô G11:

=D11 + F11

+ Sao chép công thức tại ô G11 đến vùng G12:G18.

- Hàm **HLOOKUP**(*lookup\_value*, *table\_array*, *row\_index\_num*, *range\_lookup*)

**Ví dụ:** xét lại ví dụ áp dụng hàm **VLOOKUP** ở trên nhưng BẢNG HỌC BỔNG và BẢNG TRỢ CẤP được cho như sau:

	A	B	C	D	E	F	G
1	BẢNG HỌC BỔNG		Xếp loại	01	02	03	04
2			Học bổng	100,000	70,000	50,000	30,000
3							
4	BẢNG TRỢ CẤP		Mã TC	B	A	C	
5			Tỷ lệ	50%	100%	0%	

**Giải:**

### 1) Tính cột **Học bổng**

Công thức cho ô D11 như sau:

=HLOOKUP(C11, \$D\$1:\$G\$2, 2, 1)

hoặc =HLOOKUP(C11, \$D\$1:\$G\$2, 2)

### 2) Tính cột **Trợ cấp**

Công thức cho ô F11 như sau:

=D11 \* HLOOKUP(E11, \$D\$4:\$F\$5, 2, 0)

### 3) Tính cột **Tổng cộng**

Công thức cho ô G11 như sau:

=D11 + F11